

11 Graphical User Interfaces: Brownian Motion

Previously, in Chapter 3, a simple user interface was developed that enabled easier interpretation of the output from networks of `groovyPlugAndPlay` processes. This chapter explores more complex user interfaces in conjunction with a relatively simple graphical application based upon particle movement.

The JCSP package contains an active implementation of the java AWT (Abstract Windows Toolkit). The term active is here used to mean that each AWT component, for example, button, scrollbar and canvas, has been wrapped in a process so that component events and configuration are undertaken by channel communications. This means that the active components can be connected to any process. Furthermore, the programmer does not have to write any event handling or listener methods as these are contained within the active process wrapper. The active components inherit capabilities from the basic AWT components, thus the methods and fields associated with the component can be reused and active and ordinary non-active components can be used in the same interface.

The primary benefit of the active AWT components is that processes that access the user interface can utilise their non-deterministic capabilities, thereby reflecting the unpredictable behaviour of user interfaces. The user interface has no knowledge of when, for example, a button is going to be pressed and thus either a channel communication or an alternative provides a simple method for capturing that non-deterministic behaviour.

11.1 Active AWT Widgets

The fundamental process diagram for an active widget is shown in Figure 13-1. A widget is any component available in the `java.awt` package for which an active version has been constructed. Some active widgets have been constructed that simplify the construction of user interfaces. Specific widgets may have more or less channels depending upon the functionality of the widget. All widgets have a *configure* input channel which enables the configuration of the widget at run-time. In most cases the configuration of a new widget can be defined when it is constructed, unless of course the content of the user interface is to be altered by changing the configuration of one of its widgets. For example, when a button has its associated text changed to reflect the state of the user interface. Each of the active component output channels produces data values that are related to the underlying AWT specification of that event and is specified in the `java.awt` documentation. The role of the configuration and event channels is specified in the `org.jcsp.awt` documentation and depends upon the specific component. For example, if the event arises from the pressing of an `ActiveButton` then the message communicated is the text string associated with the button. Similarly, a configuration channel message could be a text string that is to replace the current text associated with the button.




Figure 11-1 Generic Active Widget Process Diagram

11.2 The Particle System – Brownian Motion

A particle motion system¹ comprises a number of particles that move around at random. Their position is shown on a Canvas. Using Java threads and a Canvas results in a somewhat cumbersome representation of the solution because a Canvas executes in its own thread of control, which has the effect of distributing the particle control, random movement and the graphical representation throughout the classes that make up the solution.

In the parallel solution that follows these drawbacks are eliminated and the fact that a Canvas has to execute in its own thread of control is hidden from the programmer. Furthermore in this solution we shall introduce some additional capabilities. The particles will bounce off the side of the bounding Canvas. The user will be given control of the application with a button that allows them to initially start the system and then subsequently to pause and resume its operation. In addition two buttons are provided which modify the ‘temperature’ of the system. The higher the temperature the greater the random movement exhibited by the particles. The particles do not bounce off each other and that is left as an additional exercise for the interested reader.

The structure of the Brownian motion system is shown in Figure 13-2. A number of particles (Particle 0 to n) are connected to the ParticleInterface. This utilises a new form of channel called Any2One. An Any2One channel enables the connection of any number of writer processes to a single reader process. The point-to-point nature of channel communication is, however, still maintained because only one communication can proceed at a time. Communications on an Any2One are such that communication from one writer to the single reader is completed before the next writer can commence its communication. The converse is true of One2Any channels. The JCSP also includes Any2Any channels where yet again once a communication has started it behaves like a one-to-one communication.

¹ Doug Lea, Concurrent Programming in Java, Second Edition, Chapter 1,




Figure 11-2 Brownian Motion Process Network

Particles are not aware of their position relative to the sides of the bounding canvas and thus the particle may move to a position that is outwith the bounding canvas. In this case the particle's position is updated within the **ParticleInterface**. The updated position together with any change of temperature is returned to the **Particle** using the update channel. The update channel is a **One2Any** channel that permits one writer to write to any number of readers. This is not a broadcast communication because the writer can only write to one of the reader processes at any one time. Furthermore, once one of the many reader processes has committed to a communication no other reader will be able to start a communication until the writer has written to that reader process.

The **ParticleManager** is responsible for receiving inputs from the **Particle** processes; modifying their position, should the indicated position lie outside the bounding canvas; and then causing the display of the particle's position on the canvas. The **ParticleManager** is also responsible for dealing with button events from the **UserInterface** and configuring the buttons and labels within the **UserInterface**. Data is passed between the **Particle** processes and the **ParticleManager** by means of a data object that contains both positional information as well as any change to the temperature.

The **UserInterface** contains the display canvas, together with a button that is used to initially start and then subsequently used to pause and restart the system. Two further buttons are provided that are used to increase or decrease the temperature together with a **Label** that shows the current temperature value with an indication of whether the last change was up or down. The channels used between the **ParticleManager** and the **UserInterface** will be described more fully in a later section.

11.2.1 The Position Data Object

The **Position** data object, see Listing 13-1, is used to communicate data between the **Particles** and the **ParticleInterface**. **Position** implements the interface **JCSPCopy {1}**, which is defined within the **org.jcsp.groovy** package. It should be recalled that objects are passed between processes running on the same machine by means of an object reference. In some situations this could lead to the creation of a large number of newly created short-lived objects, which could lead to the calling of the automatic Java garbage collector. The calling of the garbage collector during a graphical display would interfere with the presentation. The abstract interface **JCSPCopy** defines a method called **copy()**, which can be used to generate a deep copy of an object.

```

01    class Position implements JCSPCopy {
02        def int id
03        def int lx
04        def int ly
05        def int px
06        def int py
07        def int temperature

08        def copy() {
09            def p = new Position ( id: this.id,
10                                lx: this.lx, ly: this.ly,
11                                px: this.px, py: this.py,
12                                temperature : this.temperature )
13            return p
14        }

15        def String toString () {
16            def s = "[Position-> " + id + ", " + lx + ", " + ly
17            s = s + ", " + px + ", " + py
18            s = s + ", " + temperature + " ]"
19            return s
20        }
21    }

```

Listing 11-1 The Position Data Object

Lines {2-7} define the properties of `Position`. The property `id` is the number of the `Particle`. The properties `lx` and `ly` are the newly calculated [x, y] position co-ordinates of the `Particle`. These co-ordinates may lie outside the display area. The properties `px` and `py` are the co-ordinates of the previous position of the particle. The property `temperature` maintains the current value of the temperature within the system. All the properties, apart from `id` can be altered within the `ParticleInterface`.

Lines {8-14} define the method `copy` required for the implementation of the interface `JCSPCopy`. For completeness, a `toString` method is defined {15-20} that can be used to output the contents of a `Position` object.

11.2.2 The Particle Process

The definition of the `Particle` process is shown in Listing 13-2. A `Particle` has two channels one {23}, `sendPosition`, to output its `Position` to, and the other {24}, `getPosition`, to receive updated `Positions` from the `ParticleInterface`. It should be noted that even though these channels will eventually be implemented as `Any2One` and `One2Any` channels as far as the process is concerned these are just a `ChannelOutput` and `ChannelInput` respectively. The properties `x` {25} and `y` {26} hold the initial position of the particle. A default display area of 200 pixels is presumed and thus all particles start their movement from the centre of that area. The position of the particles will be recalculated after the interval specified by `delay` {27}, which is initially set to 200 milliseconds. Each `Particle` is given a unique identification `id` {28}. The initial temperature of the system is set at 25 {29} and can range from 10 to 50.

The `run` method defines a `CSTimer` called `timer` {31} and uses the Java provided random number generator mechanism, `Random ()` {32}. The variable `p` holds the `Position` of the particle and is constructed using the initial values held within the properties passed to the process {33}.

```

22  class Particle implements CSProcess {
23      def ChannelOutput sendPosition
24      def ChannelInput getPosition
25      def int x = 100
26      def int y = 100
27      def long delay = 200
28      def int id
29      def int temperature = 25
30
31      void run() {
32          def timer = new CTimer()
33          def rng = new Random()
34          def p = new Position ( id: id, px: x, py: y, temperature: temperature )
35
36          while (true) {
37              p.lx = p.px + rng.nextInt(p.temperature) - ( p.temperature / 2 )
38              p.ly = p.py + rng.nextInt(p.temperature) - ( p.temperature / 2 )
39              p.ly = p.py + rng.nextInt(p.temperature * 2) - ( p.temperature )
40              sendPosition.write ( p )
41              p = ( (Position)getPosition.read() ).copy()
42              timer.sleep ( delay )
43          }
44      }
45  }

```

Listing 11-2 The Particle Process

The main loop of the process {35-39} requires the calculation of the new position of the particle `lx` and `ly` that are stored in the variable object `p` {35, 36}. The calculation ensures that the particle moves in a space that surrounds the current location `[px, py]` by a square with a side of size `temperature`. The position `p` is then written to the `ParticleInterface` {38}. This is a write operation that is implemented on a shared `Any2One` channel and thus the process will have to wait until any other outstanding communications have completed. An `Any2One` channel is essentially fair in that the communications are placed in a queue of such communications.

The `Particle` process behaves like a client to the `ParticleInterface`'s server. As soon as it has written its position to the `ParticleInterface` it reads the updated position information {39} from the `getPosition` channel. The `getPosition` channel is implemented by means of a `One2Any` channel and thus this client – server interaction has to be carefully considered. When the `sendPosition.write(p)` communication is completed only this `Particle` process can be in that state because only one communication is permitted on an `Any2One` channel. Hence the only process that will be in a position to undertake a read on the `getPosition` channel is this process. Hence we are assured that a `Particle` process that writes its position to `ParticleInterface` will be the one to receive its response, even though we are using shared `Any2One` and `One2Any` channels.

Finally, the `Particle` process sleeps for the delay period {40} after which the loop is repeated until the user stops the application through the user interface. The user interface will cause the `Particle` processes to stop even though they are implemented using a non-terminating while-loop.

11.2.3 The Particle Interface Process

This process, shown in Listing 13-3 is typical of any application that uses a graphical user interface in that it comprises a process that undertakes both the interaction with the user interface and the rest of the system and the a process that implements the user interface itself. These two processes are always run in parallel using communication channels to pass events and configuration information between the processes.

```

44  class ParticleInterface implements CSPProcess {
45      def ChannelInput inChannel
46      def ChannelOutput outChannel
47      def int canvassSize = 100
48      def int particles
49      def int centre
50      def int initialTemp

51      void run() {
52          def dList = new DisplayList()
53          def particleCanvas = new ActiveCanvas()
54          particleCanvas.setPaintable (dList)
55          def tempConfig = Channel.createOne2One()
56          def pauseConfig = Channel.createOne2One()
57          def uiEvents = Channel.createAny2One( new OverwriteOldestBuffer(5) )

58          def network = [ new ParticleManager ( fromParticles: inChannel,
59                                                  toParticles: outChannel,
60                                                  toUI: dList,
61                                                  fromUIButtons: uiEvents.in(),
62                                                  toUIPause: pauseConfig.out(),
63                                                  toUILabel: tempConfig.out(),
64                                                  CANVASSIZE: canvassSize,
65                                                  PARTICLES: particles,
66                                                  CENTRE: centre,
67                                                  START_TEMP: initialTemp ),

68              new UserInterface ( particleCanvas: particleCanvas,
69                                canvassSize: canvassSize,
70                                tempValueConfig: tempConfig.in(),
71                                pauseButtonConfig: pauseConfig.in(),
72                                buttonEvent: uiEvents.out() ) ]

73      new PAR ( network ).run()
74  }
75  }

```

Listing 11-3 The ParticleInterface Process

The channels `inChannel` {45} and `outChannel` {46} are used to connect this process to the `Particle` processes. Yet again this process definition does not need to be aware of the specific implementation of the channels actually used to connect the processes together. The property `canvassSize` {47} provides a default size for the display area. Similarly, properties are defined, with default values, for the number of particles {48}, the centre of the display area {49} and the initial temperature {50} of the system.

The variable `dList` {52} is of type `DisplayList`, defined within `org.jcsp.awt`. The use of `dList` will be described later. It is sufficient to note, at this stage, that it is passed as a property to the `ParticleManager` process {60}. An `ActiveCanvas`, `particleCanvas` is defined {53} and then a call to its `setPaintable()` method is made that associates it with `dList` {54}. In this manner both `ParticleManager` and `UserInterface` can access `dList`, the former directly as a property and the other indirectly through `particleCanvas` {68}. Essentially, `dList` is a shared object between the processes but the user can only modify the `dList` in `ParticleManager` directly. Therefore a `DisplayList` object has to be defined before either of the processes that access it are defined.

The `tempConfig` channel {55} is used to update the temperature display in the interface. The `pauseConfig` {56} channel is used to set the text in the START/PAUSE/RESTART button.

The `uiEvents` channel {57} passes button events from the `UserInterface` to the `ParticleManager` process. It is not possible to press two buttons at the same time hence we can use an `Any2One` channel, which simplifies processing within the `ParticleManager` process. The parameter `OverwriteOldestBuffer (5)` specifies that this channel will use a buffer of 5 elements in which, should it become full the oldest element in the buffer will be overwritten. This buffer is required because it is essential that events on this channel are always read otherwise the underlying Java event thread may block, which would also have the effect of stopping the rest of the user interface. The specified buffer will always read an input, hence ensuring that the Java event thread will not block and that another

process will always be able to read the last few events, five in this case, even if the reading process is slow.

The network {58-72} simply comprises the `ParticleManager` and `UserInterface` processes with parameters and variables passed as parameters as required to construct the process network as shown in Figure 3-2.

11.2.4 The ParticleManager Process

The properties of the `ParticleManager` process are shown in Listing 13-4. The channel connections with `Particle` processes are provided by the channels `fromParticles` {77} and `toParticles` {78}. When the system is instantiated these will be passed shared channels of type `Any2One` and `One2Any` respectively. The constant properties {79-82} respectively contain the size of the square display area (`CANVASSIZE`), number of particles (`PARTICLES`), the centre co-ordinate of the display area (`CENTRE`) and the initial value of the system temperature (`START_TEMP`). The `DisplayList` property {83}, `toUI`, provides the graphical connection between the `ParticleManager` and `UserInterface` processes. The `ChannelInput` {84} `fromUIButtons` is the channel by which button event from the user interface are communicated to `ParticleManager`. Finally, the `ChannelOutputs` `toUILabel` {85} and `toUIPause` {86} provide the means by which the temperature value and the `START`, `PAUSE` and `RESTART` button have their values changed.

```

76    class ParticleManager implements CSProcess {
77        def ChannelInput fromParticles
78        def ChannelOutput toParticles
79        def int CANVASSIZE
80        def int PARTICLES
81        def int CENTRE
82        def int START_TEMP
83        def DisplayList toUI
84        def ChannelInput fromUIButtons
85        def ChannelOutput toUILabel
86        def ChannelOutput toUIPause

```

Listing 11-4 ParticleManager Properties

The initialisation of the `ParticleManager` is shown in Listing 13-5. The variable `colourList` {88-90} contains a list of `java.awt.colors` that is used to colour the particles once they start moving. The variable `temperature` {91} is assigned the value of property `START_TEMP`.

The next part {95-106} initialises the variables that will be used by the `DisplayList` mechanism. The variable, `particleGraphics` {92} used to `set()` a `DisplayList` comprises an array of `GraphicsCommands`. The initial element of `particleGraphics` {95, 96} contains a `GraphicsCommand` that clears the display area. The remainder of `particleGraphics` comprises two elements per particle. The first element of which is a command to set the colour of the particle and the second will draw a circle of that colour with a radius of 10 pixels at the position of the particle. However for initialisation, each particle is set to the colour `BLACK` {98} and placed at the `CENTRE` {99} of the display area. This is captured in the variable `initialGraphic` {97}. The nested `for` loops {100-105} copies the `initialGraphic` into the array `particleGraphics`. Thus `particleGraphics` comprises a first command to clear the display followed by as many pairs of `GraphicsCommands` as there are particles needing to be drawn. The `DisplayList`, `toUI` is then `set()` to `particleGraphics` {106}. The manner in which the `DisplayList` is manipulated will be described later.

```

87 void run() {
88     def colourList = [ Color.BLUE, Color.GREEN,
89                       Color.RED, Color.MAGENTA,
90                       Color.CYAN, Color.YELLOW]
91     def temperature = START_TEMP
92
93     GraphicsCommand [] particleGraphics =
94         new GraphicsCommand [ 1 + (PARTICLES * 2) ]
95
96     particleGraphics[0] =
97         new GraphicsCommand.ClearRect ( 0, 0, CANVASSIZE, CANVASSIZE )
98
99     GraphicsCommand [] initialGraphic = new GraphicsCommand [ 2 ]
100    initialGraphic[0] = new GraphicsCommand.SetColor (Color.BLACK)
101    initialGraphic[1] = new GraphicsCommand.FillOval (CENTRE,CENTRE,10,10)
102
103    for ( i in 0 ..< PARTICLES ) {
104        def p = (i * 2) + 1
105        for ( j in 0 ..< 2) {
106            particleGraphics [p+j] = initialGraphic[j]
107        }
108    }
109
110    toUI.set (particleGraphics)
111
112    GraphicsCommand [] positionGraphic = new GraphicsCommand [ 2 ]
113    positionGraphic =
114        [ new GraphicsCommand.SetColor (Color.WHITE),
115          new GraphicsCommand.FillOval (CENTRE, CENTRE, 10, 10)
116        ]
117
118    def pmAlt = new ALT ( [fromUIButtons, fromParticles] )
119
120    def initTemp = " " + temperature + " "
121    toUILabel.write ( initTemp )
122
123    def direction = fromUIButtons.read()
124    while ( direction != "START" ) {
125        direction = fromUIButtons.read()
126    }
127    toUIPause.write("PAUSE")

```

Listing 11-5 ParticleManager Initialisation

The two element array `positionGraphic` {107-111} will subsequently be used to update the `DisplayList` to reflect the movement of particles. It is initialised to sensible values that will be overwritten. However it can be observed that the first element of the array contains a command to set the colour and the second causes the drawing of a circle of that colour. The `ParticleManager` process alternates over inputs from the user interface buttons, `fromUIButtons` and from the particles on channel `fromParticles` {112}. The String `initTemp` is defined to hold the initial value of `temperature` {113} surrounded by spaces. This String is then written to the label that displays this value using the channel `toUILabel` {114}.

The variable `direction` reads from the channel `fromUIButtons` {115}. A user interface button signals a button event by communicating the String that is currently displayed by the button. Recall that all the user interface buttons are connected to the same channel, `fromUIButtons`. Only the `START/PAUSE/RESTART` button has the initial value `START` and thus the process will wait until the button labelled `START` is pressed. This behaviour is captured in the while loop {116-118}, which ignores any other button events. Once `START` has been read, the button's text value is changed to `PAUSE` {119} by writing to the `toUIPause` channel. The operation of the system now commences and this is shown in Listing 13-6.


```

120     while (true) {
121         def index = pmAlt.priSelect()
122         if ( index == 0 ) { // dealing with a button event
123             direction = fromUIButtons.read()
124             if (direction == "PAUSE" ) {
125                 toUIPause.write("RESTART")
126                 direction = fromUIButtons.read()
127                 while ( direction != "RESTART" ) {
128                     direction = fromUIButtons.read()
129                 }
130                 toUIPause.write("PAUSE")
131             }
132         } else {
133             if (( direction == "Up" ) && ( temperature < 50 )) {
134                 temperature = temperature + 5
135                 def s = "+" + temperature + "+"
136                 toUILabel.write ( s )
137             }
138             else {
139                 if ( (direction == "Down" ) && ( temperature > 10 ) ) {
140                     temperature = temperature - 5
141                     def s = "-" + temperature + "-"
142                     toUILabel.write ( s )
143                 }
144                 else {
145                     }
146             }
147         }
148     }

```

Listing 11-6 ParticleManager Button Event Processing

The index of the selected alternative is obtained, with priority being given to button events {121, 122}. If the value read from the channel fromUIButtons is PAUSE {124} then it is immediately overwritten with RESTART {125}. The process then waits for the button event RESTART ignoring all other button events {126-129}. Once the system has been restarted the button is overwritten with the value PAUSE {130}.

If the value read into direction is not PAUSE then it must either be Up or Down which are the text associated with the buttons that manipulate the temperature of the system. If the up button is pressed and provided the current value of temperature is less than 50 {133} then the temperature is raised by 5 {134} and the new value of temperature is written to the interface using the channel toUILabel surrounded by + symbols {135-136}. Similarly if the Down button is pressed then the temperature is reduced by 5 provided its current value is greater than 10 and is output surrounded by - symbols {139-142}.

```

149     else { // index is 1 particle movement
150         def p = (Position) fromParticles.read()
151         if ( p.lx > CANVASSIZE ) { p.lx = (2 * CANVASSIZE) - p.lx }
152         if ( p.ly > CANVASSIZE ) { p.ly = (2 * CANVASSIZE) - p.ly }
153         if ( p.lx < 0 ) { p.lx = 0 - p.lx }
154         if ( p.ly < 0 ) { p.ly = 0 - p.ly }
155         positionGraphic [0] =
156             new GraphicsCommand.SetColor ( colourList.getAt(p.id%6 ) )
157         positionGraphic [1] = new GraphicsCommand.Filloval (p.lx,p.ly,10,10)
158         toUI.change ( positionGraphic, 1 + ( p.id * 2) )
159         p.px = p.lx
160         p.py = p.ly
161         p.temperature = temperature
162         toParticles.write(p)
163     } // index test
164 } // while
165 } // run
166 }

```

Listing 11-7 ParticleManager Particle Movement Processing

Listing 13-7 shows the processing that deals with the movement of particles. Recall that ParticleManager is behaving as a server process. Hence we would expect to see it read a client request {150}, undertake

some processing and then respond with the return value {162}. The `Position` data object is read into the variable `p` from the channel `fromParticles` {150}. The proposed location [`lx`, `ly`] of the particle is then assessed as to whether it still remains within the display area {151-154} and if not its position is adjusted assuming that the reflection from the side of the display area involves not friction or elastic compression of the particle. The value of the `PositionGraphic` array is then modified to reflect the particle's colour by taking the modulus 6 remainder of the particle's id {156} and then setting the centre of the circle to [`lx`, `ly`] {157}. This is then used to overwrite the data for this particle in the `DisplayList` parameter using the `toUI.change()` method {158}.

The description of the operation of a `DisplayList` can now be completed. An `ActiveCanvas` takes the `DisplayList` object as a parameter. Internally, the `ActiveCanvas` constructs two copies of the associated `DisplayList` array of `Graphics` commands. These copies are used to provide a double buffering mechanism; this however is hidden from the programmer. At a specified period the `ActiveCanvas` draws the current buffer on the display, while other changes are recorded in the other copy. This mechanism is repeated displaying the first buffer and recording changes in the second and then displaying the second buffer while recording changes in the first copy. The `DisplayList` is initialised by a `set` method {106}. Thereafter specific elements of the `DisplayList` can be altered using the `change` method {158}. Thus the programmer generates the effect of continually updating the display, which in fact is using a double buffering technique to smooth the repainting of the display. The user is not concerned with the repainting of the display as this handled within the `ActiveCanvas` process. Thus the `DisplayList` array of `GrahicsCommands` has an initial element that clears the display area, which is then overwritten by the sequence of `GraphicsCommands` in the array. In this manner sophisticated animation can be achieved, without having to overwrite each particle individually.

11.2.5 The UserInterface Process

The `UserInterface` process is shown in Listing 13-8. The properties of the process include the `particleCanvas`, `fromPM` {168}, the size of the canvas {169}, the two input channels, `tempValueConfig` {170} and `pauseButtonConfig` {171} used to configure the temperature value and the start button. Finally, the `buttonEvent` channel is used to output button events to the `ParticleManager` process {172}.

```

167  class UserInterface implements CSProcess {
168      def ActiveCanvas particleCanvas
169      def int canvassize
170      def ChannelInput tempValueConfig
171      def ChannelInput pauseButtonConfig
172      def ChannelOutput buttonEvent
173
174      void run() {
175          def root = new ActiveClosingFrame ("Brownian Motion Particle System")
176          def mainFrame = root.getActiveFrame()
177
178          def tempLabel = new Label ("Temperature")
179          def tempValue = new ActiveLabel (tempValueConfig)
180          tempValue.setAlignment( Label.CENTER)
181
182          def upButton = new ActiveButton (null, buttonEvent, "Up")
183          def downButton = new ActiveButton (null, buttonEvent, "Down")
184          def pauseButton = new ActiveButton ( pauseButtonConfig,
185                                              buttonEvent, "START")
186
187          def tempContainer = new Container()
188          tempContainer.setLayout ( new GridLayout ( 1, 5 ) )
189          tempContainer.add ( pauseButton )
190          tempContainer.add ( tempLabel )
191          tempContainer.add ( upButton )
192          tempContainer.add ( tempValue )
193          tempContainer.add ( downButton )
194
195          particleCanvas.setSize (canvassize, canvassize)
196          mainFrame.setLayout( new BorderLayout() )
197          mainFrame.add (particleCanvas, BorderLayout.CENTER)
198          mainFrame.add (tempContainer, BorderLayout.SOUTH)
199          mainFrame.pack()
200          mainFrame.setVisible ( true )
201
202          def network = [ root, particleCanvas,
203                        tempValue, upButton,
204                        downButton, pauseButton
205                      ]
206
207          new PAR (network).run()
208      }
209  }

```

Listing 11-8 The User Interface Process

The run method of this process comprises a list of definitions and associated method calls that instantiates the graphical user interface. First, root {174}, an ActiveClosingFrame is defined that will be used to hold the rest of the interface components. An ActiveClosingFrame is defined with the frame's title as a parameter and is not introduced by a property name because these processes are defined as Java classes and thus are constructed using the normal Java mechanism. ActiveClosingFrame is a specialisation of ActiveFrame that permits the closing of the frame using the normal window based controls. Interface components have to be added to the enclosed frame which is accessed by means of the getActiveFrameMethod() call {175}. The next part of the Listing shows the definition of the interface widgets both active and ordinary AWT non-active ones which can be mixed as required. The Label, tempLabel, which displays the text 'Temperature' is constructed {176}. An ActiveLabel called tempValue is then defined {177} with the channel tempValueConfig as its parameter. Typically, an active widget has a constructor that comprises the configuration and event channels, together with any other appropriate parameter. The alignment of the label is also specified {178}. After this the required ActiveButtons are defined {179-182}, in which the null parameter is a placeholder for the not needed *configuration* channel. The additional parameter specifies the initial text associated with the button. The pauseButton requires a configuration channel {181} because the value of the text String associated with the button changes as the application progresses.

Next a Container, tempContainer is defined {183} that holds all the components associated with the manipulation of temperature together with the pauseButton. The Container uses a GridLayout {184}.

The previously defined buttons and label are then added to the `tempContainer` {185-189}. The size of `particleCanvas` is specified {190}.

The mainframe can now be created {191-195} by specifying it to be a `BorderLayout` {191}. The `particleCanvas` and `tempContainer` are then added to the mainframe in the `CENTER` and `SOUTH` of the layout {192, 193}. The mainframe is then packed and `setVisible` {194, 195}.

Finally, a process network is constructed that comprises the root and the remaining active widgets {196-199}. The network is then run {200} and that is all that needs to be specified for the user interface requirements of this application. The event handler and listener methods normally required do not have to be written as these have been encapsulated within the active widgets, thereby simplifying the construction of the user interface.

11.2.6 Invoking the Brownian Motion System

Listing 13-9 gives the script that is required to invoke the Brownian motion system. The `Any2One` channel connect and the `One2Any` channel update are defined {203, 204}. The fundamental constants of the system are either obtained from a user interaction or defined as constants {205-208}. The empty `List` network is defined {209} to which is appended each of the `Particle` processes {211-217}. The `ParticleInterface` process is finally appended to network {218-223}. The system is then executed by running `PAR` {225}.

```

203 Any2OneChannel connect = Channel.createAny2One()
204 One2AnyChannel update = Channel.createOne2Any()

205 def CSIZE = Ask.Int ("Size of Canvas (200, 600)?": ", 200, 600)
206 def CENTRE = CSIZE / 2
207 def PARTICLES = Ask.Int ("Number of Particles (10, 200)?": ", 10, 200)
208 def INIT_TEMP = 25

209 def network = []
210 for ( i in 0..< PARTICLES ) {
211     network << new Particle ( id: i,
212                             sendPosition: connect.out(),
213                             getPosition: update.in(),
214                             x: CENTRE,
215                             y: CENTRE,
216                             temperature: INIT_TEMP )
217 }


218 network << ( new ParticleInterface ( inChannel: connect.in(),
219                                     outChannel: update.out(),
220                                     canvasSize: CSIZE,
221                                     particles: PARTICLES,
222                                     centre: CENTRE,
223                                     initialTemp: INIT_TEMP ) )
224 println "Starting Particle System"

225 new PAR ( network ).run()

```

Listing 11-9 The Script To Invoke the Brownian Motion System

A typical screen capture of the system, when it has been `PAUSED` is shown in Output 13-1. We can observe that the control button has been set to `RESTART`. The temperature is currently set at 40 and the last operation was to increase its value because it is surrounded by `+` symbols. The up and down buttons are clearly visible. The screen is derived from a system that has a canvas size of 450 pixels running 100 particles.



Output 11-1 Screen Capture of the Brownian Motion System

11.3 Summary

This chapter has described how user interfaces can be constructed very simply using the active widget concept. Of most significance is the relative simplicity of the user interface definition as it does not require the programmer to implement the event and listener methods normally required. It has introduced a standard design pattern for user interface applications in which there is a process that undertakes the processing `ParticleManager` and its associated `userInterface` process that are executed in parallel.

The concept of a `DisplayList` has been introduced which simplifies the programming of animated user interfaces based upon drawing in an `ActiveCanvas`. This in itself typifies the ease with which user interface can be constructed using active widgets because the programmer can use the parallel programming constructs to implement the interaction between user and application processes.

The design and implementation of user interfaces has become a much easier task because the user is no longer concerned with the writing of event handler and listener methods. Furthermore, the encapsulation of interface components, which run in their own thread and their associated event handler thread into a single process, makes it much easier to build the system that interacts with the interface.

11.4 Exercises

1. The Control process in the Scaling system currently updates the scaling factor according to an automatic system. Replace this with a user interface that issues the suspend communication, obtains the current scaling factor and then asks the user for the new scaling factor that is then injected into the `Scaler`. The original and scaled values should also be output to the user interface.